An effective graph can reveal trends or patterns in your data that may have otherwise remained hidden in a tabular form. SAS 9.2 introduced a new family of procedures for creation of statistical graphs ranging from simple scatter plots to paneled displays to assist in data analysis.

This family of SG procedures includes the SGPLOT, SGSCATTER and SGPANEL procedures. The SG procedures provide a concise syntax to create graphs commonly used in data analysis in many domains.

With SAS 9.3, many new features were added to make graphs easier, including Cluster Groups, Interval Box Plots, High Low plots, Attribute Maps and Annotation.

SAS 9.4 and the maintenance releases have added more features that make it really easy to customize your graphs, including setting of group attributes, splitting axis values, jittering, proportional column widths, heat maps, etc.

New features include grouped histograms and segment labels. Color and pattern groups are added to series plots to create spaghetti graphs. Splines and series plots will support thickness response in SAS 9.40M3.

The new AXISTABLE statement makes it easy to include axis aligned textual data in a graph. Marker symbols can be defined from font glyphs and image icons. The POLYGON and TEXT statements provide more flexibility for creating custom graphs.

These examples and their complete programs can be viewed at:
http://support.sas.com/rnd/datavisualization/yourGraphs/sgf/
proc sgplot data=heart nowall noborder;
  histogram bp / group=type filltype=solid
      transparency=0.5 nbins=50 name='h';
  density bp / group=type;
  xaxis display=(nolabel noline) max=250;
  yaxis display=(noline noticks) grid;
  keylegend 'h' / location=inside
      across=1 position=topright opaque;
run;

proc sgplot data=cars nowall noborder;
  vbar type / group=origin dataskin=pressed
      groupdisplay=stack seglabel
      seglabelfitpolicy=none datalabel;
  keylegend / title='' location=inside
      position=topright across=1;
  xaxis display=(nolabel);
  yaxis grid display=(noline noticks);
run;

proc sgplot data=sashelp.cars;
  vbar type / response=mpg_city stat=mean
      yaxis ranges=(0-27 43-55);
  xaxis display=(nolabel);
run;

proc sgplot data=class nowall noborder;
  styleattrs axisbreak=bracket
      axisextent=data;
  scatter x=height y=weight / group=sex
      yaxis ranges=(40-145 215-260)
      values=(40 to 260 by 20);
  keylegend / location=inside
      position=topleft;
run;
proc sgplot data=AETimelinecap
datatrmap=attrmap;
  highlow y=aeseq low=aestdate
  high=aeendate / type=bar group=aesev
  barwidth=0.8 lownlabel=aedecod
  highcap=highcap attrid=Severity;
  xaxis grid display=(nolabel);
  yaxis display=(noticks novalues nolabel) grid;
run;

proc sgplot data=QTcData;
  format week qtcweek.;
  reline 26 / axis=x;
  reline 0 30 60 / axis=y;
  vbox qtc / category=week group=drug
    groupdisplay=cluster nofill;
  xaxistable risk / class=drug
    colorgroup=drug;
  xaxis display=(nolabel) type=linear
    values=(0 2 4 8 12 16 20 24 28)
    max=29 valueshint;
  yaxis label='QTc change from baseline'
    values=(-120 to 90 by 30);
run;

proc sgplot data=heart noautolegend;
  scatter x=deathcause y=cholesterol /
    jitter filledoutlinedmarkers
    discreteoffset=-0.15
    jitterwidth=0.4
    dataskin=sheen;
  vbox cholesterol / category=deathcause
    fillattrs=(transparency=0.5)
    nooutliers discreteoffset=0.15
    boxwidth=0.2;
  xaxis display=(nolabel noticks);
  yaxis display=(noline noborder noticks);
run;

proc sgplot data=sashelp.class;
  bubble x=height y=weight size=age /
    group=sex transparency=0.2
    datalabel=age datalabelpos=center
    dataskin=sheen;
  inset 'Bubble size is prop to age' /
    position=bottomright;
  xaxis grid; yaxis grid;
  keylegend / location=inside
    position=topleft;
run;
proc sgpanel data=LuxurySedans;
  panelby origin / proportional uniscale=row
    novarname layout=columnlattice
      onepanel sort=ascmean;
  vbar make / response=msrp dataskin=gloss
    stat=mean group=origin datalabel
    categoryorder=respasc;
  colaxis display=(nolabel);
run;

proc sgplot data=sankey2 nocycleattrs;
  styleattrs datacontrastcolors=(<colors>);
  spline x=x y=y / group=id
    thickresp=thickness thickmax=84;
  highlow y=y2 low=xl high=xh /
    highlabel=hlabel lowlabel=llabel
    type=bar intervalbarwidth=16 group=id;
  highlow y=y3 low=xl high=xh /
    highlabel=hlabel lowlabel=llabel
    type=bar intervalbarwidth=84 group=id;
  text x=xlbl y=ylbl text=label;
symbol image car = "C:\Car.png";
symbol image truck = "C:\Truck.png";
  yaxis grid integer;
xaxis grid;
run;